
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 7.3.4.4.1s which is part of the POS test suite ‘POS_14wk10’ in the ‘iwd-EUTRA-B2013-03_D14wk10’ delivery.
The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

2. Table of Contents

21. Overview

2. Table of Contents
2
3. Verification Test Summary
2
4. Corrections required for test case 7.3.4.4.1s
3
     Change 1 – Correction to local function ‘fl_TC_7_3_4_4_GNSS_TestBody’
3
     Change 2 – Correction to function ‘f_POS_AgnssProvideAssistanceData_UEA’
7
     Change 3 – Adding new templates to support GNSS Error check
11
     Change 4 – Correction of AGNSS templates
12
6. Execution Log Files
17
6.1 Nvidia T4i UE
17
References
17

3. Verification Test Summary

Test Case:
TC_7_3_4_4_1s
Test Group:
POS\7
ATS Version:
iwd-EUTRA-B2013-03_D14wk10
System Simulator used:
R&S CMW500  
UE used:
Nvidia T4i 
Verification Status:
PASS


4. Corrections required for test case 7.3.4.4.1s
Change 1 – Correction to local function ‘fl_TC_7_3_4_4_GNSS_TestBody’
	Function name
	fl_TC_7_3_4_4_GNSS_TestBody()

	Reason for change
	1.   In step 3 and step 5 additional information must be present in the indication because it had been sent in the request.
2.   In step 5, the constraint ‘cr_LppProvideLocationInformation_UEA’ does not allow for checking presence of Signalling Measurement Information; i.e. whether the measurement was successful or whether GNSS Error was sent as LPP message. As a consequence, non-conformant UEs would also PASS the test case implementation. 

	Summary of change
	1.   Considered parameter ‘cr_RoutingInfo(tsc_POS_RoutingId)’ in the receive constraints for “LPP Request Assistance Data“ in step 3  resp. the receive constraint for “LPP PROVIDE LOCATION INFORMATION” in step 5.
2.   In step 5, a check for presence of Signalling Measurement Information is introduced to prevent that “GNSS Error” is interpreted as Signalling Measurement Information. New constraint ‘cr_LppProvideLocationInformation_Agnss_UEA’ is introduced (see change 3 below). 

	TTCN module
	POS\7\EUTRA_Positioning_Testcases.ttcn

	MCC160 Comment
	


Before change

...

  function fl_TC_7_3_4_4_GNSS_TestBody (PositioningSystemList_Type p_PosSystemListId,
                                        bitstring p_Gnss_Ids) runs on EUTRA_POS_PTC
  {
    var ProvideAssistanceData_r9_IEs v_ProvideAssistData;
    var LPP_Message v_LppMessage;
    var octetstring v_Octetstring;
    var octetstring v_EncodedLppMessage;
    var SRB_COMMON_IND v_ReceivedAsp;
    var RequestAssistanceData_r9_IEs v_RequestAssistData;
    var TransactionNumber v_TransactionNumber := 0; // To store Transaction number sent by the UE in LPP Request Assistance Data
    //@siclog "Step 1" siclog@
    //Clear stored GNSS assistance data
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_EUTRA_RESETUEPOSITIONINGSTOREDINFORMATION (tsc_AgnssUEPosTech))));
    //@siclog "Steps 1b, 1c, 1d" siclog@
    f_POS_CapabilityTransferProcedure (); //@sic R5-134909 sic@
    //@siclog "Step 2" siclog@
    //SS sends LPP REQUEST LOCATION INFORMATION
    v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppRequestLocationInfo (tsc_POS_TransactionNumber,
                                                                          locationMeasurementsRequired,
                                                                          tsc_ResponseTimeDef,
                                                                          cs_AgnssRequestLocationInfo (p_Gnss_Ids,
                                                                                                       true),
                                                                          omit,
                                                                          omit));
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                           cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                           cs_RoutingInfo(tsc_POS_RoutingId)))));
    //@siclog "Step 3" siclog@
    //Receive LPP Request Assistance Data
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  omit))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppReqAssistanceDataAgnss_Ack);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 3");
    // Get Transaction number sent by the UE
    v_TransactionNumber := v_LppMessage.transactionID.transactionNumber;
    v_RequestAssistData := v_LppMessage.lpp_MessageBody.c1.requestAssistanceData.criticalExtensions.c1.requestAssistanceData_r9;
    // Retrieve requested data from Satellite simulator and store it in v_ProvideAssistData
    v_ProvideAssistData := f_POS_RetrieveEUTRA_DataSubtest (f_POS_PosSystem2Subtest (p_PosSystemListId), ////@sic R5s130627 sic@
                                                            false); //UE assisted
    //@siclog "Step 4" siclog@
    //SS sends LPP PROVIDE ASSISTANCE DATA
    if (isvalue(v_LppMessage.acknowledgement)) {// UE requested ack
      v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppProvideAssistData (v_TransactionNumber,
                                                                          targetDevice,
                                                                          f_POS_AgnssProvideAssistanceData_UEA(f_POS_PosSystem2Subtest(p_PosSystemListId), //@sic R5s130627 sic@
                                                                                                               v_RequestAssistData,
                                                                                                               v_ProvideAssistData),
                                                                          omit,
                                                                          cs_Acknowledgement (false, v_LppMessage.sequenceNumber)));
      SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo(tsc_POS_RoutingId)))));
    }
    else {// UE did not request ack
      v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppProvideAssistData (v_TransactionNumber,
                                                                          targetDevice,
                                                                          f_POS_AgnssProvideAssistanceData_UEA(f_POS_PosSystem2Subtest(p_PosSystemListId), //@sic R5s130627 sic@
                                                                                                               v_RequestAssistData,
                                                                                                               v_ProvideAssistData),
                                                                          omit));
      SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo(tsc_POS_RoutingId)))));
    }
    //@siclog "Step 5" siclog@
    //Receive LPP PROVIDE LOCATION INFORMATION
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  omit))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppProvideLocationInformation_UEA (tsc_POS_TransactionNumber));
    if (isvalue (v_LppMessage.lpp_MessageBody.c1.provideLocationInformation.criticalExtensions.c1.provideLocationInformation_r9.a_gnss_ProvideLocationInformation)) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 5");
    }
    else {
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "A-GNSS Location Information not present");
    }
    //@siclog "Step 5a" siclog@
    //SS sends LPP ACKNOWLEDGEMENT if requested by the UE at step 5
    f_POS_SendLppAck (eutra_Cell1,
                      tsc_POS_RoutingId,
                      v_LppMessage);
  }
... 
After change

...

  function fl_TC_7_3_4_4_GNSS_TestBody (PositioningSystemList_Type p_PosSystemListId,
                                        bitstring p_Gnss_Ids) runs on EUTRA_POS_PTC
  {
    var ProvideAssistanceData_r9_IEs v_ProvideAssistData;
    var LPP_Message v_LppMessage;
    var octetstring v_Octetstring;
    var octetstring v_EncodedLppMessage;
    var SRB_COMMON_IND v_ReceivedAsp;
    var RequestAssistanceData_r9_IEs v_RequestAssistData;
    var TransactionNumber v_TransactionNumber := 0; // To store Transaction number sent by the UE in LPP Request Assistance Data
    //@siclog "Step 1" siclog@
    //Clear stored GNSS assistance data
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_EUTRA_RESETUEPOSITIONINGSTOREDINFORMATION (tsc_AgnssUEPosTech))));
    //@siclog "Steps 1b, 1c, 1d" siclog@
    f_POS_CapabilityTransferProcedure (); //@sic R5-134909 sic@
    //@siclog "Step 2" siclog@
    //SS sends LPP REQUEST LOCATION INFORMATION
    v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppRequestLocationInfo (tsc_POS_TransactionNumber,
                                                                          locationMeasurementsRequired,
                                                                          tsc_ResponseTimeDef,
                                                                          cs_AgnssRequestLocationInfo (p_Gnss_Ids,
                                                                                                       true),
                                                                          omit,
                                                                          omit));
    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                tsc_SRB2,
                                cs_TimingInfo_Now,
                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                               cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                           cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                           cs_RoutingInfo(tsc_POS_RoutingId)))));
    //@siclog "Step 3" siclog@
    //Receive LPP Request Assistance Data
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  cr_RoutingInfo(tsc_POS_RoutingId)))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppReqAssistanceDataAgnss_Ack);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 3");
    // Get Transaction number sent by the UE
    v_TransactionNumber := v_LppMessage.transactionID.transactionNumber;
    v_RequestAssistData := v_LppMessage.lpp_MessageBody.c1.requestAssistanceData.criticalExtensions.c1.requestAssistanceData_r9;
    // Retrieve requested data from Satellite simulator and store it in v_ProvideAssistData
    v_ProvideAssistData := f_POS_RetrieveEUTRA_DataSubtest (f_POS_PosSystem2Subtest (p_PosSystemListId), ////@sic R5s130627 sic@
                                                            false); //UE assisted
    //@siclog "Step 4" siclog@
    //SS sends LPP PROVIDE ASSISTANCE DATA
    if (isvalue(v_LppMessage.acknowledgement)) {// UE requested ack
      v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppProvideAssistData (v_TransactionNumber,
                                                                          targetDevice,
                                                                          f_POS_AgnssProvideAssistanceData_UEA(f_POS_PosSystem2Subtest(p_PosSystemListId), //@sic R5s130627 sic@
                                                                                                               v_RequestAssistData,
                                                                                                               v_ProvideAssistData),
                                                                          omit,
                                                                          cs_Acknowledgement (false, v_LppMessage.sequenceNumber)));
      SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo(tsc_POS_RoutingId)))));
    }
    else {// UE did not request ack
      v_EncodedLppMessage := f_POS_EncodeLppMessage (cs_LppProvideAssistData (v_TransactionNumber,
                                                                          targetDevice,
                                                                          f_POS_AgnssProvideAssistanceData_UEA(f_POS_PosSystem2Subtest(p_PosSystemListId), //@sic R5s130627 sic@
                                                                                                               v_RequestAssistData,
                                                                                                               v_ProvideAssistData),
                                                                          omit));
      SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,
                                  tsc_SRB2,
                                  cs_TimingInfo_Now,
                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_DL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                             cs_GenericMsgContainer(v_EncodedLppMessage),
                                                                             cs_RoutingInfo(tsc_POS_RoutingId)))));
    }
    //@siclog "Step 5" siclog@
    //Receive LPP PROVIDE LOCATION INFORMATION
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_UL_GENERIC_NAS_TRANSPORT(tsc_LPP_MessageContainer,
                                                                                  cr_RoutingInfo(tsc_POS_RoutingId)))))
    -> value v_ReceivedAsp;
    v_Octetstring := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_GENERIC_NAS_TRANSPORT.genericMessageContainer.genericMsgContainerValue;
    v_LppMessage := f_POS_LppMessage_DecodeAndMatch (v_Octetstring, cr_LppProvideLocationInformation_Agnss_UEA (tsc_POS_TransactionNumber));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 7.3.4.4 Step 5");
    //@siclog "Step 5a" siclog@
    //SS sends LPP ACKNOWLEDGEMENT if requested by the UE at step 5
    f_POS_SendLppAck (eutra_Cell1,
                      tsc_POS_RoutingId,
                      v_LppMessage);
  }
... 
Change 2 – Correction to function ‘f_POS_AgnssProvideAssistanceData_UEA’
	Function name
	f_POS_AgnssProvideAssistanceData_UEA ()

	Reason for change
	In function f_POS_AgnssProvideAssistanceData_UEA the “GNSS Time Model” related IE and the “GNSS UTC Model” related IE are removed from the default contents in variable v_A_gnss_ProvideAssistanceData regardless of executed sub-test. However, these two IEs should only be omitted in case of sub-test 4. See TS 37.571-5 clause 6.1.3.3 and 6.1.3.4.

	Summary of change
	Added an “if” statement to remove the 2x IEs only in case of sub-test 4.

	TTCN module
	POS\EUTRA\EUTRA_Positioning_Functions.ttcn

	MCC160 Comment
	


Before change

...

  //----------------------------------------------------------------------------
  /*
   * @desc      To build the data to be sent to the UE based on what the UE requested in p_RequestAssitData. UE-Assisted case. @sic R5s130627 sic@
   * @param     p_Subtest
   * @param     p_RequestAssistData
   * @param     p_ProvideAssistData
   * @return    A_GNSS_ProvideAssistanceData
   * @status
   */
  function f_POS_AgnssProvideAssistanceData_UEA(integer p_Subtest,
                                                RequestAssistanceData_r9_IEs p_RequestAssistData,
                                                ProvideAssistanceData_r9_IEs p_ProvideAssistData) return A_GNSS_ProvideAssistanceData
  {
    var template A_GNSS_ProvideAssistanceData v_A_gnss_ProvideAssistanceData;
    select (p_Subtest) { //Initialize with default data
      case (1) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Gps_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Gps_NoAA(p_ProvideAssistData);
        }
      }
      case (2) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Glonass_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Glonass_NoAA(p_ProvideAssistData);
        }
      }
      case (3) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Galileo_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Galileo_NoAA(p_ProvideAssistData);
        }
      }
      case (4) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_GpsAndGlonass_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_GpsAndGlonass_NoAA(p_ProvideAssistData);
        }
      }
    }
    //Remove those IE not requested by the UE
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceTimeReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceLocationReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceLocation := omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_IonosphericModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_IonosphericModel:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_TimeModelsReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_TimeModels := omit;
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_TimeModels := omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_NavigationModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_NavigationModel:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_NavigationModel:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AcquisitionAssistanceReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AcquisitionAssistance:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_AcquisitionAssistance:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AlmanacReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_Almanac:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_UTCModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_UTC_Model := omit;
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_UTC_Model := omit;
    }
    if (not ispresent (p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AuxiliaryInformationReq)){
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_AuxiliaryInformation := omit;
      }
    }
    return (valueof(v_A_gnss_ProvideAssistanceData));
  }
... 
After change

...

  //----------------------------------------------------------------------------
  /*
   * @desc      To build the data to be sent to the UE based on what the UE requested in p_RequestAssitData. UE-Assisted case. @sic R5s130627 sic@
   * @param     p_Subtest
   * @param     p_RequestAssistData
   * @param     p_ProvideAssistData
   * @return    A_GNSS_ProvideAssistanceData
   * @status
   */
  function f_POS_AgnssProvideAssistanceData_UEA(integer p_Subtest,
                                                RequestAssistanceData_r9_IEs p_RequestAssistData,
                                                ProvideAssistanceData_r9_IEs p_ProvideAssistData) return A_GNSS_ProvideAssistanceData
  {
    var template A_GNSS_ProvideAssistanceData v_A_gnss_ProvideAssistanceData;
    select (p_Subtest) { //Initialize with default data
      case (1) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Gps_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Gps_NoAA(p_ProvideAssistData);
        }
      }
      case (2) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Glonass_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Glonass_NoAA(p_ProvideAssistData);
        }
      }
      case (3) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Galileo_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_Galileo_NoAA(p_ProvideAssistData);
        }
      }
      case (4) {
        if (pc_GNSS_AA) {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_GpsAndGlonass_AA(p_ProvideAssistData);
        }
        else {
          v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEA_GpsAndGlonass_NoAA(p_ProvideAssistData);
        }
      }
    }
    //Remove those IE not requested by the UE
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceTimeReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceLocationReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceLocation := omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_IonosphericModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_IonosphericModel:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_TimeModelsReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_TimeModels := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_TimeModels := omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_NavigationModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_NavigationModel:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_NavigationModel:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AcquisitionAssistanceReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AcquisitionAssistance:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_AcquisitionAssistance:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AlmanacReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_Almanac:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_UTCModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_UTC_Model := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_UTC_Model := omit;
      }
    }
    if (not ispresent (p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AuxiliaryInformationReq)){
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_AuxiliaryInformation := omit;
      }
    }
    return (valueof(v_A_gnss_ProvideAssistanceData));
  }
...  
Change 3 – Adding new templates to support GNSS Error check 
	Template name
	cr_A_GNSS_ProvideLocationInformation_LI (),
cr_A_GNSS_ProvideLocationInformation_SMI (),
cr_LppProvideLocationInformation_Agnss_UEA ()

	Reason for change
	Template name of ‘cr_A_GNSS_ProvideLocationInformation’ did not specify the IE to be checked. It is designed for checking Location Information only. 

See also change 1.2 above which uses the ‘cr_LppProvideLocationInformation_Agnss_UEA’.

	Summary of change
	Renaming previous (+ adding new) template(s) to distinguish between checking Location Information (LI) and Signalling Measurement Information (SMI).
New contraint ‘cr_LppProvideLocationInformation_Agnss_UEA’ added similar to ‘cr_LppProvideLocationInformation_Agnss_UEB ()’.

	TTCN module
	POS\EUTRA\EUTRA_LPP_Templates.ttcn

	MCC160 Comment
	


Before change

...
  template (present) A_GNSS_ProvideLocationInformation cr_A_GNSS_ProvideLocationInformation :=
  { /* @status    APPROVED (POS) */
    gnss_SignalMeasurementInformation := omit,
    gnss_LocationInformation := {
      measurementReferenceTime := ?,
      agnss_List := ?
    },
    gnss_Error := omit
  };
...

After change

...
  template (present) A_GNSS_ProvideLocationInformation cr_A_GNSS_ProvideLocationInformation_LI :=
  { /* @status    APPROVED (POS) */
    gnss_SignalMeasurementInformation := omit,
    gnss_LocationInformation := {
      measurementReferenceTime := ?,
      agnss_List := ?
    },
    gnss_Error := omit
  };
  template (present) A_GNSS_ProvideLocationInformation cr_A_GNSS_ProvideLocationInformation_SMI :=
  { /* @status    APPROVED (POS) */
    gnss_SignalMeasurementInformation := {
      measurementReferenceTime := ?,
      gnss_MeasurementList := ?
    },
    gnss_LocationInformation := omit,
    gnss_Error := omit
  };
  template (present) LPP_Message cr_LppProvideLocationInformation_Agnss_UEA(TransactionNumber p_TransactionNumber) :=
  /* @status    APPROVED (POS) */
    cr_LppProvideLocationInformation_Common(p_TransactionNumber, cr_ProvideLocationInformation_r9_IEs(omit, cr_A_GNSS_ProvideLocationInformation_SMI, omit, omit)); //@sic R5-131099 sic@
... 
Change 4 – Correction of AGNSS templates 
	Template name
	cs_A_GNSS_RequestAssistanceData_UEA_Gps_NoAA(),
cs_AgnssProvideAssistData_UEA_Gps_AA(),
cs_AgnssProvideAssistData_UEA_Gps_NoAA(),
cs_GNSS_AcquisitionAssistance

	Reason for change
	1.   As per TS 37.571-5 section 6.1.3.4 (=  Default Assistance Data for TS 37.571-2 subclause 7), the satellite ID {1, 11, 17, 20, 23, 28} shall be used for sub-test 1. See Table “GNSS-NavModelSatelliteElement: sub-test 1” for details.
2.   ‘Confidence_r10’ is missing in template ‘cs_GNSS_AcquisitionAssistance’.

	Summary of change
	1.   Replaced satellite ID {8, 11, 17, 19, 27, 28} with satellite ID {1, 11, 17, 20, 23, 28}.

2.   Added r10 parameter.

	TTCN module
	POS\EUTRA\EUTRA_LPP_Templates.ttcn

	MCC160 Comment
	


Before change

...

  // UE ASSISTED --------------------------------------------------------------------
  template (value) A_GNSS_RequestAssistanceData cs_A_GNSS_RequestAssistanceData_UEA_Gps_AA :=
  { // UE supports Acquisition Assistance
    /* @status    APPROVED (POS) */
    gnss_CommonAssistDataReq := cs_GNSS_CommonAssistDataReq(cs_GNSS_ReferenceTimeReq({cs_GNSS_ID_Gps}, true, false)),
    gnss_GenericAssistDataReq := {
      cs_GNSS_GenericAssistDataReqElement(cs_GNSS_ID_Gps,
                                          -,
                                          -,
                                          cs_GNSS_AcquisitionAssistanceReq(0))  // GPS L1 C/A
    }
  };
  template (value) A_GNSS_RequestAssistanceData cs_A_GNSS_RequestAssistanceData_UEA_Gps_NoAA :=
  { // UE does not support Acquisition Assistance
    /* @status    APPROVED (POS) */
    gnss_CommonAssistDataReq := cs_GNSS_CommonAssistDataReq(cs_GNSS_ReferenceTimeReq({cs_GNSS_ID_Gps}, true, false),
                                                            cs_GNSS_ReferenceLocationReq),
    gnss_GenericAssistDataReq := {
      cs_GNSS_GenericAssistDataReqElement(cs_GNSS_ID_Gps,
                                          -,
                                          cs_GNSS_NavigationModelReq_ReqNavList(tsc_SvID_8_11_17_19_27_28, true),
                                          -,
                                          cs_GNSS_AlmanacReq)
    }
  };
...

  //==============================================================================
  // AGNSS PROVIDE ASSISTANCE DATA - UE ASSISTED
  // According to 37.571-2, 5.4.1.1
  //==============================================================================
  template (value) A_GNSS_ProvideAssistanceData cs_AgnssProvideAssistData_UEA_Gps_AA (ProvideAssistanceData_r9_IEs p_ProvideAssistData) :=
  { // Acquisition Assistance Supported by the UE
    /* @status    APPROVED (POS) */
    gnss_CommonAssistData := {
      gnss_ReferenceTime := cs_GNSS_ReferenceTime_Gps(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime, {8, 11, 17, 19, 27, 28}),
      gnss_ReferenceLocation := omit,
      gnss_IonosphericModel := omit,
      gnss_EarthOrientationParameters := omit
    },
    gnss_GenericAssistData := {
      {
        gnss_ID := cs_GNSS_ID_Gps,
        sbas_ID := omit,
        gnss_TimeModels := omit,
        gnss_DifferentialCorrections := omit,
        gnss_NavigationModel := omit,
        gnss_RealTimeIntegrity := omit,
        gnss_DataBitAssistance := omit,
        gnss_AcquisitionAssistance := cs_GNSS_AcquisitionAssistance(cs_GNSS_SignalID(0), // GPS L1 C/A
                                                                    fl_Build_cs_GNSS_AcquisitionAssistList(tsc_IntList_1_11_17_20_23_28, // {1, 11, 17, 20, 23, 28},
                                                                                                           p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AcquisitionAssistance.gnss_AcquisitionAssistList)),
        gnss_Almanac := omit,
        gnss_UTC_Model := omit,
        gnss_AuxiliaryInformation := fl_Build_cs_GNSS_AuxiliaryInformation( tsc_IntList_8_11_17_19_27_28, //{8, 11, 17, 19, 27, 28},
                                                                            p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation)
      }
    },
    gnss_Error := omit
  };
  template (value) A_GNSS_ProvideAssistanceData cs_AgnssProvideAssistData_UEA_Gps_NoAA (ProvideAssistanceData_r9_IEs p_ProvideAssistData) :=
  { // Acquisition Assistance NOT Supported by the UE
    /* @status    APPROVED (POS) */
    gnss_CommonAssistData := {
      gnss_ReferenceTime := cs_GNSS_ReferenceTime_Gps(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime, {8, 11, 17, 19, 27, 28}),
      gnss_ReferenceLocation := cs_GNSS_ReferenceLocation(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceLocation),
      gnss_IonosphericModel := omit,
      gnss_EarthOrientationParameters := omit
    },
    gnss_GenericAssistData := {
      {
        gnss_ID := cs_GNSS_ID_Gps,
        sbas_ID := omit,
        gnss_TimeModels := omit,
        gnss_DifferentialCorrections := omit,
        gnss_NavigationModel := cs_NavigationModel_Gps(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_NavigationModel),
        gnss_RealTimeIntegrity := omit,
        gnss_DataBitAssistance := omit,
        gnss_AcquisitionAssistance := omit,
        gnss_Almanac := {
          weekNumber := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.weekNumber,
          toa := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.toa,
          ioda := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.ioda,
          completeAlmanacProvided := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.completeAlmanacProvided,
          gnss_AlmanacList := fl_Build_cs_GNSS_AlmanacList(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.gnss_AlmanacList)
        },
        gnss_UTC_Model := omit,
        gnss_AuxiliaryInformation := fl_Build_cs_GNSS_AuxiliaryInformation( tsc_IntList_8_11_17_19_27_28, //{8, 11, 17, 19, 27, 28},
                                                                            p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation)
      }
    },
    gnss_Error := omit
  };
...
/*
   * @desc      Builds IE GNSS_AcquisitionAssistance for all satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedAcquisitionAssistList
   * @return    template (value) GNSS_AcquisitionAssistList
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_AcquisitionAssistList(IntegerList_Type p_SatelliteIdList,
                                                  GNSS_AcquisitionAssistList p_ProvidedAcquisitionAssistList) return template (value) GNSS_AcquisitionAssistList
  {
    var template (value) GNSS_AcquisitionAssistList v_GNSS_AcquisitionAssistList;
    var integer i;
    for (i := 0; i < 6; i:= i + 1) {
      v_GNSS_AcquisitionAssistList[i] := cs_GNSS_AcquisitionAssistElement(p_SatelliteIdList[i], p_ProvidedAcquisitionAssistList[i]);
    }
    return v_GNSS_AcquisitionAssistList;
  }
  template (value) GNSS_AcquisitionAssistance cs_GNSS_AcquisitionAssistance(template (value) GNSS_SignalID p_SignalID,
                                                                            template (value) GNSS_AcquisitionAssistList p_AcquisitionAssistList) :=
  { /* @status    APPROVED (POS) */
    gnss_SignalID := p_SignalID,
    gnss_AcquisitionAssistList := p_AcquisitionAssistList,
    confidence_r10 := omit    /* @sic R5s130195 BASELINE MOVING 2013 sic@ */
  };
...
After change

...

  // UE ASSISTED --------------------------------------------------------------------
  template (value) A_GNSS_RequestAssistanceData cs_A_GNSS_RequestAssistanceData_UEA_Gps_AA :=
  { // UE supports Acquisition Assistance
    /* @status    APPROVED (POS) */
    gnss_CommonAssistDataReq := cs_GNSS_CommonAssistDataReq(cs_GNSS_ReferenceTimeReq({cs_GNSS_ID_Gps}, true, false)),
    gnss_GenericAssistDataReq := {
      cs_GNSS_GenericAssistDataReqElement(cs_GNSS_ID_Gps,
                                          -,
                                          -,
                                          cs_GNSS_AcquisitionAssistanceReq(0))  // GPS L1 C/A
    }
  };
  template (value) A_GNSS_RequestAssistanceData cs_A_GNSS_RequestAssistanceData_UEA_Gps_NoAA :=
  { // UE does not support Acquisition Assistance
    /* @status    APPROVED (POS) */
    gnss_CommonAssistDataReq := cs_GNSS_CommonAssistDataReq(cs_GNSS_ReferenceTimeReq({cs_GNSS_ID_Gps}, true, false),
                                                            cs_GNSS_ReferenceLocationReq),
    gnss_GenericAssistDataReq := {
      cs_GNSS_GenericAssistDataReqElement(cs_GNSS_ID_Gps,
                                          -,
                                          cs_GNSS_NavigationModelReq_ReqNavList(tsc_SvID_1_11_17_20_23_28, true),
                                          -,
                                          cs_GNSS_AlmanacReq)
    }
  };
...

  //==============================================================================
  // AGNSS PROVIDE ASSISTANCE DATA - UE ASSISTED
  // According to 37.571-2, 5.4.1.1
  //==============================================================================
  template (value) A_GNSS_ProvideAssistanceData cs_AgnssProvideAssistData_UEA_Gps_AA (ProvideAssistanceData_r9_IEs p_ProvideAssistData) :=
  { // Acquisition Assistance Supported by the UE
    /* @status    APPROVED (POS) */
    gnss_CommonAssistData := {
      gnss_ReferenceTime := cs_GNSS_ReferenceTime_Gps(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime, {1, 11, 17, 20, 23, 28}),
      gnss_ReferenceLocation := omit,
      gnss_IonosphericModel := omit,
      gnss_EarthOrientationParameters := omit
    },
    gnss_GenericAssistData := {
      {
        gnss_ID := cs_GNSS_ID_Gps,
        sbas_ID := omit,
        gnss_TimeModels := omit,
        gnss_DifferentialCorrections := omit,
        gnss_NavigationModel := omit,
        gnss_RealTimeIntegrity := omit,
        gnss_DataBitAssistance := omit,
        gnss_AcquisitionAssistance := cs_GNSS_AcquisitionAssistance(cs_GNSS_SignalID(0), // GPS L1 C/A
                                                                    fl_Build_cs_GNSS_AcquisitionAssistList(tsc_IntList_1_11_17_20_23_28, // {1, 11, 17, 20, 23, 28},
                                                                                                           p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AcquisitionAssistance.gnss_AcquisitionAssistList),
                                                                    p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AcquisitionAssistance.confidence_r10),
        gnss_Almanac := omit,
        gnss_UTC_Model := omit,
        gnss_AuxiliaryInformation := fl_Build_cs_GNSS_AuxiliaryInformation( tsc_IntList_1_11_17_20_23_28, // {1, 11, 17, 20, 23, 28},
                                                                            p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation)
      }
    },
    gnss_Error := omit
  };
  template (value) A_GNSS_ProvideAssistanceData cs_AgnssProvideAssistData_UEA_Gps_NoAA (ProvideAssistanceData_r9_IEs p_ProvideAssistData) :=
  { // Acquisition Assistance NOT Supported by the UE
    /* @status    APPROVED (POS) */
    gnss_CommonAssistData := {
      gnss_ReferenceTime := cs_GNSS_ReferenceTime_Gps(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime, {1, 11, 17, 20, 23, 28}),
      gnss_ReferenceLocation := cs_GNSS_ReferenceLocation(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceLocation),
      gnss_IonosphericModel := omit,
      gnss_EarthOrientationParameters := omit
    },
    gnss_GenericAssistData := {
      {
        gnss_ID := cs_GNSS_ID_Gps,
        sbas_ID := omit,
        gnss_TimeModels := omit,
        gnss_DifferentialCorrections := omit,
        gnss_NavigationModel := cs_NavigationModel_Gps(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_NavigationModel),
        gnss_RealTimeIntegrity := omit,
        gnss_DataBitAssistance := omit,
        gnss_AcquisitionAssistance := omit,
        gnss_Almanac := {
          weekNumber := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.weekNumber,
          toa := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.toa,
          ioda := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.ioda,
          completeAlmanacProvided := p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.completeAlmanacProvided,
          gnss_AlmanacList := fl_Build_cs_GNSS_AlmanacList(p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_Almanac.gnss_AlmanacList)
        },
        gnss_UTC_Model := omit,
        gnss_AuxiliaryInformation := fl_Build_cs_GNSS_AuxiliaryInformation( tsc_IntList_1_11_17_20_23_28, //{1, 11, 17, 20, 23, 28},
                                                                            p_ProvideAssistData.a_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation)
      }
    },
    gnss_Error := omit
  };
... 
  /*
   * @desc      Builds IE GNSS_AcquisitionAssistance for all satellites in the list
   * @param     p_SatelliteIdList
   * @param     p_ProvidedAcquisitionAssistList
   * @return    template (value) GNSS_AcquisitionAssistList
   * @status    APPROVED (POS)
   */
  function fl_Build_cs_GNSS_AcquisitionAssistList(IntegerList_Type p_SatelliteIdList,
                                                  GNSS_AcquisitionAssistList p_ProvidedAcquisitionAssistList) return template (value) GNSS_AcquisitionAssistList
  {
    var template (value) GNSS_AcquisitionAssistList v_GNSS_AcquisitionAssistList;
    var integer i;
    for (i := 0; i < 6; i:= i + 1) {
      v_GNSS_AcquisitionAssistList[i] := cs_GNSS_AcquisitionAssistElement(p_SatelliteIdList[i], p_ProvidedAcquisitionAssistList[i]);
    }
    return v_GNSS_AcquisitionAssistList;
  }
  template (value) GNSS_AcquisitionAssistance cs_GNSS_AcquisitionAssistance(template (value) GNSS_SignalID p_SignalID,
                                                                            template (value) GNSS_AcquisitionAssistList p_AcquisitionAssistList,
                                                                            integer p_confidence_r10) :=
  {
    gnss_SignalID := p_SignalID,
    gnss_AcquisitionAssistList := p_AcquisitionAssistList,
    confidence_r10 := p_confidence_r10    /* @sic R5s130195 BASELINE MOVING 2013 sic@ */
  };
... 
6. Execution Log Files

6.1 Nvidia T4i UE 

The Nvidia T4i UE passed this test case on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 3. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
Rohde-Schwarz\tc_7_3_4_4_1s_LTE_UE_POS_Nvidia.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
References

	[1]
	R5s140254:    Supporting information for agreement of EUTRA UE positioning TC 7.3.4.4.1s. 

                         -  text format execution log files.

	
	


