
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-EUTRA-B2013-03_D14k10’.

3. Corrections

Change 1 – Correction to function ‘f_POS_AgnssProvideAssistanceData_UEB’
	Function name
	f_POS_AgnssProvideAssistanceData_UEB ()

	Reason for change
	In function f_POS_AgnssProvideAssistanceData_UEB the “GNSS Time Model” related IE and the “GNSS UTC Model” related IE are removed from the default contents in variable v_A_gnss_ProvideAssistanceData regardless of executed sub-test. 

However, these two IEs should only be omitted in case of sub-test 4. 

	Summary of change
	Added an “if” statement to remove the 2x IEs only in case of sub-test 4. 

	TTCN module
	POS\EUTRA\EUTRA_Positioning_Functions.ttcn

	MCC160 Comment
	


Before:

	…
  //----------------------------------------------------------------------------
  /*
   * @desc      To build the data to be sent to the UE based on what the UE requested in p_RequestAssitData. UE-Based case.
   * @param     p_Subtest
   * @param     p_RequestAssistData
   * @param     p_ProvideAssistData
   * @return    A_GNSS_ProvideAssistanceData
   * @status    APPROVED (POS)
   */
  function f_POS_AgnssProvideAssistanceData_UEB(integer p_Subtest,
                                                RequestAssistanceData_r9_IEs p_RequestAssistData,
                                                ProvideAssistanceData_r9_IEs p_ProvideAssistData) return A_GNSS_ProvideAssistanceData
  { /* @sic R5s130627 sic@ */
    var template A_GNSS_ProvideAssistanceData v_A_gnss_ProvideAssistanceData;
    select (p_Subtest) { //Initialize with default data
      case (1) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_Gps(p_ProvideAssistData);
      }
      case (2) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_Glonass(p_ProvideAssistData);
      }
      case (3) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_Galileo(p_ProvideAssistData);
      }
      case (4) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_GpsAndGlonass(p_ProvideAssistData);
      }
    }
    //Remove those IE not requested by the UE
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceTimeReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceLocationReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceLocation := omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_IonosphericModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_IonosphericModel:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_TimeModelsReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_TimeModels := omit;
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_TimeModels := omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_NavigationModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_NavigationModel:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_NavigationModel:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_UTCModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_UTC_Model := omit;
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_UTC_Model := omit;
    }
    if (not ispresent (p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AuxiliaryInformationReq)){
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_AuxiliaryInformation := omit;
      }
    }
    return (valueof(v_A_gnss_ProvideAssistanceData));
  }
  //----------------------------------------------------------------------------
…


After:

	…

  //----------------------------------------------------------------------------
  /*
   * @desc      To build the data to be sent to the UE based on what the UE requested in p_RequestAssitData. UE-Based case.
   * @param     p_Subtest
   * @param     p_RequestAssistData
   * @param     p_ProvideAssistData
   * @return    A_GNSS_ProvideAssistanceData
   * @status    APPROVED (POS)
   */
  function f_POS_AgnssProvideAssistanceData_UEB(integer p_Subtest,
                                                RequestAssistanceData_r9_IEs p_RequestAssistData,
                                                ProvideAssistanceData_r9_IEs p_ProvideAssistData) return A_GNSS_ProvideAssistanceData
  { /* @sic R5s130627 sic@ */
    var template A_GNSS_ProvideAssistanceData v_A_gnss_ProvideAssistanceData;
    select (p_Subtest) { //Initialize with default data
      case (1) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_Gps(p_ProvideAssistData);
      }
      case (2) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_Glonass(p_ProvideAssistData);
      }
      case (3) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_Galileo(p_ProvideAssistData);
      }
      case (4) {
        v_A_gnss_ProvideAssistanceData := cs_AgnssProvideAssistData_UEB_GpsAndGlonass(p_ProvideAssistData);
      }
    }
    //Remove those IE not requested by the UE
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceTimeReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceTime:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_ReferenceLocationReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_ReferenceLocation := omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_CommonAssistDataReq.gnss_IonosphericModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_CommonAssistData.gnss_IonosphericModel:= omit;
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_TimeModelsReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_TimeModels := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_TimeModels := omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_NavigationModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_NavigationModel:= omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_NavigationModel:= omit;
      }
    }
    if (not ispresent(p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_UTCModelReq)) {
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_UTC_Model := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_UTC_Model := omit;
      }
    }
    if (not ispresent (p_RequestAssistData.a_gnss_RequestAssistanceData.gnss_GenericAssistDataReq[0].gnss_AuxiliaryInformationReq)){
      v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[0].gnss_AuxiliaryInformation := omit;
      if (p_Subtest == 4) {
        v_A_gnss_ProvideAssistanceData.gnss_GenericAssistData[1].gnss_AuxiliaryInformation := omit;
      }
    }
    return (valueof(v_A_gnss_ProvideAssistanceData));
  }
  //----------------------------------------------------------------------------

…


